..short programs...

Systematic Savings Revisited

Stu Denenberg

Referring to "Systematic Savings" on page 132 of the Nov/Dec '77 *Creative Computing*, the fancy mathematics formula masks what is happening.

Why not just do the calculation as a person would do with a hand calculator? We could begin with the simpler problem of calculating compound interest and then slightly modify that procedure to do systematic investments. For example, the Basic program for compound interest is:

```
5 PRINT "AT END OF YEAR","BALANCE"
10 READ P,R,N
20 FOR I=1 TO N
30 P=P+P*R
40 PRINT I,,P
50 NEXT I
60 STOP
70 DATA 100,.1,10
80 END
```

```
RUN
AT END OF YEAR          BALANCE
 1                       110
 2                       121
 3                       133.1
 4                       146.41
 5                       161.051
 6                       177.156
 7                       194.872
 8                       214.359
 9                       235.795
 10                      259.374
```

Note especially that Line 30 is *not* P= P* (I + R); instead it stresses what we *actually* do when we calculate interest — namely multiply the principal by the interest rate and then add that back onto the principal to give the new principal.

Now the program to do systematic savings is exactly the same as the one for compound interest but instead of letting our 100 bucks lay around all lonely while it's compounding, we keep feeding in lumps of $100 at the end of each year so now the program looks like:

```
5 PRINT "AT END OF YEAR     AMOUNT INVESTED     TOTAL ACCUMULATED"
10 READ N,C,R
20 P=C
30 FOR I=1 TO N
40 P=P+P*R
50 PRINT TAB(5);I;TAB(25);I*C;TAB(45);P
60 P=P+C
70 NEXT I
80 STOP
90 DATA 10,100,.1
100 END
```

```
RUN
AT END OF YEAR     AMOUNT INVESTED     TOTAL ACCUMULATED
      1                 100                 110
      2                 200                 231
      3                 300                 364.1
      4                 400                 510.51
      5                 500                 671.561
      6                 600                 848.717
      7                 700                 1043.59
      8                 800                 1257.95
      9                 900                 1493.74
      10                1000                1753.12
```

C is the constant amount we save each year. Line 60 is the only *real* difference between the two programs and it shows how we add in the constant savings to our principal each year.

Dr. Stuart Denenberg
Dept. of Computer Science
SUNY
Plattsburgh, NY 12901

Compound Interest

If $1000 is deposited in a savings account paying 8%. interest compounded n times a year, then this will accumulate to

$$\$1000(1 + .08/n)^n$$

at the end of one year assuming that no deposits or withdrawals are made.

n	8% Compounded	Accumulation at end of one year. (Rounded to nearest cent)	
1	Yearly	$1000(1+.08/1)^1 =	$1080.00
2	Semiannually	$1000(1+.08/2)^2 =	$1081.60
4	Quarterly	$1000(1+.08/4)^4 =	$1082.43
12	Monthly	$1000(1+.08/12)^12 =	$1083.00
365	Daily	$1000(1+.08/365)^365 =	$1083.28
8760	Hourly	$1000(1+.08/8760)^8760 =	$1083.29
525,600	Every minute	$1000(1+.08/525600)^525600 =	$1083.29
31,536,000	Every second	$1000(1+.08/31536000)^31536000=	$1083.29

Hardly worth quibbling over hours, minutes, and seconds.